

用字母标记法进行SSR测验的 BASIC程序设计

翟保平

(吉林省农科院植保所)

摘 要

在方差分析中,用字母标记法表示SSR测验的结果,较其它方法具有简洁明了的特点。本文利用BASIC语言中的字符串截取函数及并置功能,设计编制了用字母标记法进行SSR测验的计算机程序,可在IBMPC、长城0520、PB—700和PC—1500等微机上运行。

对试验结果做方差分析时,若F测验显著,尚须进行多重比较,以明确各个平均数之间的差异显著性。其常用的方法之一即新复极差测验,也称SSR测验。对于SSR测验的结果,往往采用字母标记法表示,尤其在处理平均数较多时,这种形式具有简洁明了的特点。已有的方差分析电算程序中(如文献^[1]),对SSR测验结果的表示方法,一般只依次输出各对平均数的差数,并以星号表示其显著与否,分析起来很不方便。为此,本文根据文献^[2]所述的字母标记法,用BASIC语言编制了相应的计算机程序,并在IBMPC、长城0520、PB—700和PC—1500等微机上运行通过。稍做修改即可在其它机型上运行。将程序开头的输入语句段做些改动,可做为方差分析程序中的一个子程序调用。

一、标记方法

先将平均数从大到小排序,在最大平均数后面标上字母a(指5%平准),并与以下各平均数相比。凡差数不显著的均标上a,直至某一与之相差显著的平均数则标以b;再以该标有b的平均数与其上方各个比它大的平均数比,凡不显著的也一律标以b;再以标有b的最大平均数为准,与其下面各未标记的平均数比,凡不显著的同样一律标以b,直至某一与之相差显著的平均数则标以c。如此重复,直到最小的平均数有了标记字母为止。对于1%平准,则标以大写英文字母。这样,各平均数间凡有一相同标记字母的即差异不显著,凡具不同标记字母的即差异显著。

二、程序设计

(一)运行条件:

已知试验处理数、平均数的标准误、各处理平均数和各SSR α 值。

(二)最小显著极差 $LSR_{0.05}$ 和 $LSR_{0.01}$ 的计算:

$$LSR_{0.05} = SE \times SSR_{0.05}$$

$$LSR_{0.01} = SE \times SSR_{0.01}$$

其中,SSR $_{0.05}$ 和SSR $_{0.01}$ 的值可按组内均方所具有的自由度从Duncan's表查得。

(三) 变量说明:

1. **简单变量:** K—处理数; D—平均数的极差; E—比较结束标志; SE—平均数的标准误; SR— SSR_{α} 值; P、Q—某两级差间所包含的平均数个数; Z—每次下行比较时最大平均数的处理号; T—上行循环的次数; V—上行循环时各极差均不显著的标志。

2. **数组变量:** K\$(K)—各处理名称; X(K)—各处理平均数; R(K)—各 LSR_{α} 值; A\$(K)、B\$(K)、C\$(K)、D\$(K)、E\$(K)—存放标记字母的字符串。

(四) 各程序段的主要功能:

20—80 输入试验处理数、平均数的标准误和各平均数,并将各均数从大到小排序。

90—250 分别做 5% 和 1% 显著水平下的多重比较。

100 输入各 SSR_{α} 值, 计算并打印输出相应的 LSR_{α} 值。

105 输入各标记字母。

110—235 循环比较。其中, A\$(K) 和 C\$(K) 用以暂存标记字母, E\$(K) 和 D\$(K) 分别存放 5% 平准和 1% 平准的标记字符串。用字符串运算的方法将每次循环标记的字母合并, 同时, 以字符串截取函数保证同一平均数后面不重标相同的字母。

170 判断有无必要上行比较。

180—230 上行比较。其中, 190 行判断极差显著与否, 显著则跳出循环, 从标有同一字母的最大平均数起重做下行比较。

240—250 5% 平准比较完毕, 将结果存放于 E\$(K) 中, 并将其它字符变量置空, 将结束标志冲零。

260—290 依次打印输出处理名称、平均数、5% 和 1% 显著水平比较结果。

三、运行实例

以文献^[2]中的例题上机运行, 以便检验运行结果。

将处理数、平均数的标准误、各处理名称及其平均数、 $SSR_{0.05}$ 值和小写标记字母 (a—h)、 $SSR_{0.01}$ 值和大写标记字母 (A—H) 分别依次写入 DATA 语句。

例 5.8 六种施氮法植株含氮量的 SSR 测验。其中, $K=6$, $SE=0.1041$, $\bar{x}_1 - \bar{x}_6$ 分别为 12.52, 13.76, 13.12, 10.66, 14.48 和 13.64。按自由度 $\nu=24$ 查附表 7 得 $p=2, 3, \dots, 6$ 时的 $SSR_{0.05}$ 和 $SSR_{0.01}$ 值, 输入运行, 结果如下:

$$\alpha=0.05$$

$$LSR(2)=0.303972$$

$$LSR(3)=0.319537$$

$$LSR(4)=0.327915$$

$$LSR(5)=0.335202$$

$$LSR(6)=0.341448$$

$$\alpha=0.01$$

$$LSR(2)=0.412236$$

$$LSR(3)=0.430974$$

$$LSR(4)=0.441384$$

$$LSR(5)=0.450753$$

$$LSR(6)=0.456999$$

5	14.43	a	A
2	13.76	b	B
6	13.64	b	B
3	13.12	c	C
1	12.52	d	D
4	10.66	e	E

例 5.12 三种肥料施于三种土壤的小麦产量的 SSR 测验。其中, $K=9$, $SE=0.554$, $\bar{x}_1 - \bar{x}_9$ 分别为 20.9, 18.3, 17.2, 12.8, 12.9, 13.7, 13.4, 13.7 和 13.5。按 $\nu=18$ 查得 $SSR_{0.05}$ 和 $SSR_{0.01}$ 值, 运

行结果如下:

	$\alpha=0.05$	
LSR(2)=1.64538	LSR(3)=1.72848	LSR(4)=1.77834
LSR(5)=1.81158	LSR(6)=1.83928	LSR(7)=1.8559
LSR(8)=1.86698	LSR(9)=1.87806	
	$\alpha=0.01$	
LSR(2)=2.25478	LSR(3)=2.36553	LSR(4)=2.42652
LSR(5)=2.47084	LSR(6)=2.50962	LSR(7)=2.54286
LSR(8)=2.57056	LSR(9)=2.59272	

A1B1	20.9	a	A
A1B2	18.3	b	B
A1B3	17.2	b	B
A2B3	13.7	c	C
A3B2	13.7	c	C
A3B3	13.5	c	C
A3B1	13.4	c	C
A2B2	12.9	c	C
A2B1	12.8	c	C

例 7.3 小麦品比试验产量结果的 SSR 测验。其中, $K=8$, $SE=0.74$, $\bar{x}_1-\bar{x}_2$ 分别为 10.7, 12.4, 11.4, 10.0, 14.2, 10.8, 11.9 和 11.4。按 $\nu=14$ 查得 $SSR_{0.05}$ 和 $SSR_{0.01}$ 值。运行结果为:

	$\alpha=0.05$	
LSR(2)=2.2422	LSR(3)=2.3532	LSR(4)=2.4198
LSR(5)=2.4642	LSR(6)=2.4938	LSR(7)=2.5086
LSR(8)=2.5234		
	$\alpha=0.01$	
LSR(2)=3.1154	LSR(3)=3.2708	LSR(4)=3.3670
LSR(5)=3.4262	LSR(6)=3.4780	LSR(7)=3.5372
LSR(8)=3.5742		

E	14.2	a	A
B	12.4	ab	AB
G	11.9	ab	AB
H	11.4	b	AB
C	11.4	b	AB
F	10.8	b	AB
A	10.7	b	AB
D	10.0	b	B

四、源程序清单

```

10 REM Program of alphabetical mark in SSR test.....SSR, AM1
20 CLEAR: READ K, SE
30 DIM X(K), K$(K), R(K), A$(K), B$(8), C$(K)D$(K), E$(K)
40 FOR I=1 TO K: READ K$(I), X(I): NEXT I
50 FOR I=1 TO K-1: FOR J=I+1 TO K
60 IF X(I) >= X(J) THEN 80
70 T=X(J): X(I)=X(J): X(J)=T: T$=K$(I): K$(I)=K$(J): K$(J)=T$
80 NEXT J, I: PRINT
90 FOR N=0 TO 1: IF N=0 THEN ALPHA=0.05 ELSE ALPHA=0.01
100 PRINT TAB(34); CHR$(224); " = "; ALPHA: PRINT
110 FOR I=2 TO K: READ SR: R(I)=SE*SR*PRINT "LSR( "; I; ") = "; R(I), : NEXT I

```

```

120 PRINT : FOR I=1 TO 8 : READ B$(I) : NEXT I
130 D$(1)=B$(1) : Z=1 : T=0
140 FOR I=Z TO K-1 : P=1 : Q=1
150 FOR J=I+1 TO K : IF J=K THEN E=-1
160 D=X(I)-X(J) : P=P+1
170 IF D>R(P) THEN 200 ELSE A$(J)=B$(I-T)
180 IF RIGHT$(D$(J), 1)=A$(J) THEN 190 ELSE D$(J)=D$(J)+A$(J)
190 NEXT J : GOTO 300
200 A$(J)=B$(I+1-T)
210 IF RIGHT$(D$(J), 1)=A$(J) THEN 220 ELSE D$(J)=D$(J)+A$(J)
220 IF J-1=I THEN 290
230 FOR L=J-1 TO I+1 STEP -1
240 D=X(L)-X(J) : Q=Q+1 : IF D>R(Q) THEN Z=L+1 : T=T+1+V : V=0 : GOTO 140
250 C$(L)=A$(J)
260 IF RIGHT$(D$(L), 1)=C$(L) THEN 230
270 D$(L)=D$(L)+C$(L)
280 NEXT L : V=1
290 IF E=-1 THEN 360 ELSE NEXT I
300 IF N=1 THEN 320
310 FOR M=1 TO K : E$(M)=D$(M) : A$(M)=" " : C$(M)=" " : D$(M)=" " : NEXT M
320 PRINT : E=0 : NEXT N
330 PRINT TAB(18) : FOR I=0 TO 39 : PRINT "# "; : NEXT I : PRINT : PRINT
340 FOR I=1 TO K : PRINT TAB(21); K$(I); TAB(29); X(I);
350 PRINT TAB(43); E$(I); TAB(52); D$(I) : NEXT I : PRINT
360 PRINT TAB(18) : FOR I=0 TO 39 : PRINT "# "; : NEXT I : PRINT
370 END
380 DATA 9, .554, A1B1, 20.9, A1B2, 18.3, A1B3, 17.2, A2B1, 12.8, A2B2, 12.9
390 DATA A2B3, 13.7, A3B1, 13.4, A3B2, 13.7, A3B3, 13.5
400 DATA 2.97, 3.12, 3.21, 3.27, 3.32, 3.35, 3.37, 3.39, a, b, c, d, e, f, g, h
410 DATA 4.07, 4.27, 4.38, 4.46, 4.53, 4.59, 4.64, 4.68, A, B, C, D, E, F, G, H

```

参 考 文 献

- [1] 翟婉莹、佟立伟:《农业试验统计BASIC程序》, 辽宁科学技术出版社, 1987年。
- [2] 南京农学院:《田间试验和统计方法》, 农业出版社, 1979年, 90—148。

A BASIC PROGRAM OF ALPHABETICAL MARK METHOD FOR SSR TEST

Zhai Baoping

(The Institute of Plant Protection,
Jilin Academy of Agricultural Sciences)

ABSTRACT

A BASIC program for alphabetical mark method of SSR test was designed to make multiple comparisons following the Duncan's. The string operators, that is concatenation and string functions, were used in it. The program run well on microcomputers such as IBM PC, Great Wall, PC-1500, PB-700 etc.